Pensieve header: Implementing the scapegoated VdV algebra of http://drorbn.net/bbs/show?shot=-VanDerVeen-160731-121550.jpg.

```
ϵ /: ϵ^n_ /; n > 1 := 0;
PBWBasis = {c, u, w};
B[U@c, U@w] = -(B[U@w, U@c] = U@w);
B[U@u, U@c] = -(B[U@c, U@u] = U@u);
B[U@w, U@u] = -(B[U@u, U@w] = (t - 1) U[] - ϵ U[u, w] + 2 ϵ t U[c]);

UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[]; UU[l_, r___] := U[l] ** UU[r];
U_i_[ℰ_] := ℰ /. {t → t_i, u_U :> UU @@ Replace[u, x_ :> x_i, 1]};

B[x_, x_] = 0;
B[U[(x_)_i_], U[(y_)_i_]] := U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;

x_ ≤ y_ := OrderedQ[{x, y}]; x_ ≺ y_ := ! OrderedQ[{y, x}];
Simp[ℰ_] := Collect[ℰ, _U, Expand];

Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := (*U[x]**U[y] =*) If[x ≺ y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := (*U[x] ** U[y1,yy] =*)
  If[x ≤ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := (*U[xx,xn]**U[yy] =*) U@xx ** (U@xn ** U@yy);

σ[i_, j_][ℰ_] := ℰ /. {b_i → b_j, t_i → t_j, c_i → c_j, u_i → u_j, w_i → w_j};

m[i_, j_][ℰ_] :=
  Simp[ℰ /. x_U :> DeleteCases[x, _j] ** U @@ Cases[x, y_j :> y_i] /. {b_j → b_i, t_j → t_i}];
m[i_, j_, k_][ℰ_] := ℰ // m[i, j] // σ[i, k];

Δ[i_, j_, k_][ℰ_] := Simp[ℰ /. {
     a_. U[] :> (a /. {b_i → b_j + b_k, t_i → t_j t_k}) U[],
     a_. x_U :> (a /. {b_i → b_j + b_k, t_i → t_j t_k}) NonCommutativeMultiply @@ (x /. {
           c_i → U@c_j + U@c_k, u_i → t_k U@u_j + ϵ t_k UU[c_k, u_j] + U@u_k,
           w_i → U@w_j + ϵ UU[c_k, w_j] + U@w_k, y_1_ :> U@y_1
          })
    }]
```

```
S[i_][ℰ_] := Simp[ℰ /. {a_. x_U ⧴ (a /. {b_i → -b_i, t_i → t_i^-1}) S[i][x]}];
S[i_][U[]] = U[];
S[i_][U[y_j_, more___]] /; i ≠ j := U[y_j] ** S[i][U[more]];
S[i_][U[c_i_, more___]] := S[i][U[more]] ** (-U@c_i);
S[i_][U[u_i_, more___]] := S[i][U[more]] ** (-t_i^-1 U@u_i + Expand[ϵ t_i^-1 UU[u_i, c_i]]);
S[i_][U[w_i_, more___]] := S[i][U[more]] ** (-U@w_i + Expand[ϵ UU[w_i, c_i]]);

CoUnit[i_][ℰ_] := Simp[ℰ /. {a_. x_U ⧴ (a /. {b_i → 0, t_i → 1}) CoUnit[i][x]}] ;
CoUnit[i_][U[]] = U[];
CoUnit[i_][U[y_j_, more___]] /; i ≠ j := U[y_j] ** CoUnit[i][U[more]];
CoUnit[i_][U[y_i_, more___]] := 0;

LBasis[n_Integer] := LBasis[Range[n]];
LBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
       Table[{U@c_i, U@u_i, U@w_i, ϵ U@c_i, ϵ U@u_i, ϵ U@w_i}, {i, S}],
       Table[{U[u_i, w_j], ϵ U[u_i, w_j],
         ϵ U @@ Sort@{c_i, c_j}, ϵ U[c_i, u_j], ϵ U[c_i, w_j]}, {i, S}, {j, S}],
       Table[{ϵ U[c_i, u_j, w_k], ϵ U @@ Sort@{u_i, u_j, w_k}, ϵ U @@ Sort@{u_i, w_j, w_k}},
        {i, S}, {j, S}, {k, S}],
       Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]

BLBasis[n_Integer] := BLBasis[Range[n]];
BLBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
       Table[{U@c_i, ϵ U@c_i}, {i, S}],
       Table[{U[u_i, w_j], ϵ U[u_i, w_j], ϵ U @@ Sort@{c_i, c_j}}, {i, S}, {j, S}],
       Table[{ϵ U[c_i, u_j, w_k]}, {i, S}, {j, S}, {k, S}],
       Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]

UExp[ℰ_, n_] := Module[{t = U[], k}, U[] + Sum[(t = t ** ℰ)/k!, {k, n}]] // Simp

ToDegree[n_][ℰ_] := Simp[ℰ /. ħ → 1] /.
   {ϵ → ħ ϵ, b_i_ ⧴ ħ b_i, t_i_ ⧴ e^(ħ b_i), b → ħ b, t ⧴ e^(ħ b), x_U ⧴ ħ^Count[x,u|u_] x} /.
  a_. x_U ⧴ Normal[Series[a, {ħ, 0, n}]] * x
```

Testing AS and Jacobi

```
BLBasis[2]
```

{U[], ∈ U[], U[c_1], U[c_2], ∈ U[c_1], ∈ U[c_2], U[u_1, w_1], U[u_1, w_2], U[u_2, w_1], U[u_2, w_2], ∈ U[c_1, c_1], ∈ U[c_1, c_2], ∈ U[c_2, c_2], ∈ U[u_1, w_1], ∈ U[u_1, w_2], ∈ U[u_2, w_1], ∈ U[u_2, w_2], ∈ U[c_1, u_1, w_1], ∈ U[c_1, u_1, w_2], ∈ U[c_1, u_2, w_1], ∈ U[c_1, u_2, w_2], ∈ U[c_2, u_1, w_1], ∈ U[c_2, u_1, w_2], ∈ U[c_2, u_2, w_1], ∈ U[c_2, u_2, w_2], ∈ U[u_1, u_1, w_1, w_1], ∈ U[u_1, u_1, w_1, w_2], ∈ U[u_1, u_1, w_2, w_2], ∈ U[u_1, u_2, w_1, w_1], ∈ U[u_1, u_2, w_1, w_2], ∈ U[u_1, u_2, w_2, w_2], ∈ U[u_2, u_2, w_1, w_1], ∈ U[u_2, u_2, w_1, w_2], ∈ U[u_2, u_2, w_2, w_2]}

```
bas = BLBasis[2]; Table[B[x, y] + B[y, x], {x, bas}, {y, bas}] // Flatten // Union
```

{0}

```
bas = BLBasis[2]; Timing[
 Table[
    {x, y, z} = xyz;
    Simp[B[B[x, y], z] + B[B[y, z], x] + B[B[z, x], y]],
    {xyz, Subsets[bas, {3}]}
   ] // Flatten // Union
]
```

{26.375, {0}}

Testing Yang-Baxter

```
R[i_, j_, d_] := Simp[Module[{n, p},
     Sum[
      Sum[n = p - m;
```

$$\frac{b_i^m}{m!\,n!}\,UU[u_i^n, c_j^m, w_j^n] + \frac{\epsilon\, m\, b_i^{m-1}}{m!\,n!}\,UU[u_i^n, c_i, c_j^m, w_j^n] - \text{If}[p<d, 1, 0]\,\frac{\epsilon\, n\,(n-1)\, b_i^m}{4\, m!\, n!}\,UU[u_i^n, c_j^m, w_j^n],$$

```
       {m, 0, p}],
      {p, 0, d}
     ]
    ]] // ToDegree[d]
```

```
R[1, 2, 1]
```

U[] + ℏ b_1 U[c_2] + ∈ ℏ U[c_1, c_2] + ℏ U[u_1, w_2]

```
R3[d_] := (ToDegree[d][R[1, 2, d] ** R[1, 3, d]] ** R[2, 3, d]) -
   (ToDegree[d][R[2, 3, d] ** R[1, 3, d]] ** R[1, 2, d])
```

```
R3[1] // Simp // ToDegree[1]
```

0

```
R3[2] // ToDegree[2]
```

0

```
(R3[3] // ToDegree[3]) /. ϵ → 0
```

0

```
R3[4] // ToDegree[4]
```

0

```
R3[5] // ToDegree[5]
```

0